
This is a Maple worksheet/tutorial on Numerical Methods for approximating solutions of Differential 
Equations (DEs).  Along with expanding your toolbox, we shall explore the power of Maple for gaining insight
into DEs.  Our mission is to solve the first order DE:

d
 dx

 y x( ) = f x, y( )  ----------------->  Differential Equation

y x0( ) = y0    ----------------------->   Initial Condition

In lecture we discussed analytic methods for solving seperable DEs, and DEs that can be put into standard form.  
For example...





The analytic solution (i.e. the pencil and paper solution) using methods from section 6.11 are outlined for the two 
examples above.  The basic Maple command to solve DEs is 'dsolve'.  See the Maple manual for a more detailed 
account of the syntax associated with this Maple command.  Here are some maple commands to sole the 
examples above.



> restart:

> with(DEtools,DEplot):
with(plots):

Warning, the name changecoords has been redefined

> eqn1:=diff(y(x),x)=-2*x*y(x)/(1+x^2);  #Enter in the DE for first example 
to be solved
init:=y(0)=2;

eqn1 := d
 dx

 y x( ) = - 2 x y x( )

1 + x 2

init := y 0( ) = 2

> dsolve({eqn1,init},y(x));

y x( ) = 2

1 + x 2

> y1:=rhs(%); #Assigns the Right Hand Side (RHS) of the output from previous
command to variable y

y1 := 2

1 + x 2

The Maple solution matches the hand-written solution found above (and found it almost immediately).  We could 
indeed check that the Maple solution solves the DE...

> diff(y1,x); #LHS of DE

- 4 x

1 + x 2( )
2

> simplify(-2*x*y1/(1+x^2)); #RHS of DE

- 4 x

1 + x 2( )
2

> subs(x=0,y1);  #Check Initial Condition
2

The LHS and the RHS of the DE match and the initial condition is satisfied according to Maple.  We could also 
plot the solution as well...

> plot(y1,x=-5..5,color=black);

> eqn2:=diff(y(x),x)=-3*x^2*y(x)+6*x^2; #Enter in the DE for second example 
to be solved

eqn2 := d
 dx

 y x( ) = -3 x 2 y x( ) + 6 x 2

> dsolve(eqn2,y(x));

y x( ) = 2 + e -x 3( ) _C1

Again the Maple solution matches the hand-written example.  The answer above is a family of solutions 
dependent on the constant of integration (_C1).  This family of solutions can be reduced to single solution when 
provided an initial condition.  For example, let y(0)=0. Maple can be made to solve for the constant of integration 
by doing the following... 

> solve(subs(x=0,rhs(%))=0,_C1);

- 2

e 0

> simplify(%);
-2



> subs(_C1=%,rhs(%%%)); 
                         

2 - 2 e -x 3( )

> y2:=%;

y2 := 2 - 2 e -x 3( )

Now that the solution has been found and assigned, it can be plotted

> plot(y2,x=-1..5);

That was a good warm-up.  Notice the power of Maple for gaining insight into the problem solving process.  
Now onto Numerical Methods for approximating solutions to DEs.  The integration techniques offered for 
solving seperable equations and utilizing integrating factors are powerful tools that cater to specific classes of 
DEs.  However, there are classes of DEs that do not have analytic solutions for which numerical methods offer 
insight and approximate solutions to DEs. As such, we will explore Maple's DEtools and DEplot  with regards to
slope fields and create a subroutine to implement Euler's Method for approximating the solution to a DE.

SLOPE FIELDS



When solving 
d

 dx
 y x( ) = f x, y( ) with initial condition y x0( ) = y0 slope fields are a useful tool for visualizing the 

solution to a DE.  By form of the first order DE, the function f describes the slope of the tangent line of a solution
at a point (x,y) in the plane.  By drawing several short line segments (tangent line segments) with slope described
by f at several points in the plane, one can get a general idea of how the solutions act in the plane.  The process of
drawing enough short line segments to visualize the action of solutions in the plane would be quite tedious.  
Maple's DEtools library contains several useful commands for plotting and visualizing solutions of DEs.  Of 
particular use is the command DEplot.  Let's have Maple construct the slope fields for the previous examples (see
Maple manual for more information on these commands).  

> DEplot(eqn1,y(x),x=-5..5,y=-5..5,color=blue);

> DEplot(eqn1,y(x),x=-5..5,{[0,2]},y=-5..5,color=blue,linecolor=red);

Notice how the direction field indicates the general shape of the solution for the first example.  Also, notice how 
the syntax was changed in DEplot in the second version to include the solution going through the point (0,2).  
We could include another solution going through the point (0,4) by doing the following

> DEplot(eqn1,y(x),x=-5..5,{[0,2],[0,4]},y=-5..5,color=blue,linecolor=[red,g
reen]);

For the second example the slope field looks like

> DEplot(eqn2,y(x),x=-1..5,y=-3..3,color=blue);

> DEplot(eqn2,y(x),x=-1..5,{[0,0]},y=-3..3,color=blue,linecolor=red);
plot(y2,x=-1..5);

Once the condition that the solution must go through the origin (as done when dsolve was used above) is added 
to the syntax, notice the difference between the plot found using dsolve and the plot of the solution in the slope 
field.  Remember, when using slope fields or numerical methods in general, the output from these only produce 
approximations (so be careful).

EULER'S METHOD

Recall that the linearization of a differentiable function at a point  (the linearization is the equation of the tangent 
line at the point) is a good approximation to the function near the point.  When  

d
 dx

 y x( ) = f x, y( ) with initial 

condition y x0( ) = y0 can not be solved analytically, numerical methods, which approximate the solution at a set of 

points in the solution's domain, become useful.  The basic idea behind Euler's Method is to split the domain into 
equally spaced points (with uniform spacing dx between the points), and then patch together a string of 
linearizations starting at (x0, y0) to approximate the solution on a specified domain.  Euler's method provides a 

sequence of points to approximate y(x) at the equally spaced points starting from (x0, y0) using 

yn + 1 = yn + f xn, yn( ) dx with yn = y xn( ) (i.e.Euler's method defines yn + 1 recursively in terms of the known 



information from (xn, yn)).  Keep in mind that the method only provides approximates to the solution whose 

accuracy depends on dx and the number of steps used (n).  Errors may grow if n is too large (or dx is too big).  
See section 6.12 in text for a more detailed description of the method.  Here we utilize a Maple subroutine to 
program Euler's method for the first example above.

> f:=(x,y)->-2*x*y/(1+x^2); #f for the first example above

f := x, y( ) ® - 2 x y

1 + x 2

> euler_approx:=proc(f,x_start,y_start,dx,n_total)
local x,y,Y,i;

x[1]:=x_start;
y[1]:=y_start;

for i from 1 to n_total do
  y[i+1]:=y[i]+f(x[i],y[i])*dx;
  x[i+1]:=x[1]+i*dx;
end do;

Y:=[seq([x[k],y[k]],k=1..n_total)];
return(Y);
end;

euler_approx := proc f, x_start, y_start, dx, n_total( )
local x, y, Y, i;

x[1] := x_start;
y[1] := y_start;
for i to n_total do y[i + 1] := y[i] + f x[i], y[i]( )*dx; x[i + 1] := x[1] + i*dx; end do;
Y := seq x[k], y[k][ ], k = 1 .. n_total( )[ ];
return Y;

end proc;

> y_approx:=euler_approx(f,0,2,0.01,200):

> pointplot(%,symbol=circle);

> EulerSoln:=pointplot(%%,symbol=circle):
ExactSoln:=plot(y1,x=0..2,color=red,thickness=2):
display(EulerSoln,ExactSoln);

The Euler method approximation matches well against the exact solution.  This concludes this Maple 
worksheet/tutorial.  Continue to tend to your toolboxes...


